
Parallel Cable III Emulator for 
the XSA Board 

November 18, 2001 (Version 1.0) Application Note by D. Vanden Bout 

Summary 

This application note shows how to configure the XC9572XL CPLD on the XSA Board so its parallel port interface 
emulates the functions of the Xilinx Parallel Cable III.  This lets you use the Xilinx WebPACK JTAG tools with the 
XSA Board through its simple 25-wire downloading cable. 

 
Why Emulate the  Parallel Cable III? 

Xilinx WebPACK software contains tools for 
downloading and testing SpartanII FPGAs through 
their their JTAG interface.  One of the ways these 
tools access the FPGA is through a Parallel Cable III 
connected from the FPGA to the parallel port of a PC.  
A schematic for the Parallel Cable III  (henceforth 
referred to as PCBLIII) is shown in Figure 1. 

The SpartanII FPGA on the XSA Board is accessed 
from the PC parallel port through a simple 25-wire 
cable that connects to an XC9572XL CPLD on the 
XSA Board.  The CPLD can be programmed to pass 
signals from the parallel port to and from the JTAG 
pins of the SpartanII device.  The XSA Board is 
supplied with a default CPLD configuration that lets 
you download bitstreams to the SpartanII using the 
GXSLOAD utility provided by XESS. 

This application note describes an alternate circuit 
that allows the XSA Board CPLD to emulate the 
JTAG functions of the PCBLIII.  By loading this circuit 
into the CPLD, you can use all the Xilinx downloading 
and testing tools with the XSA Board through the 
simple downloading cable provided by XESS. 

VHDL for the Parallel Cable III Emulator 

Listing 1 shows the VHDL code for the PCBLIII 
emulator that is programmed into the XC9572XL 
CPLD on the XSA Board.  This interface provides two 
functions: 

• It transfers configuration bitstreams from the PC 
to the SpartanII FPGA using the JTAG interface. 

• After the SpartanII FPGA is configured and its 
DONE pin goes high, the JTAG interface is used 
to readback and/or test the FPGA. 

How the VHDL implements these functions is 
described below. 

Line 38 disables the Flash RAM until after the FPGA 
is configured.  The clock signal from the 
programmable oscillator is passed to a dedicated 
clock pin of the SpartanII on line 41. 

Line 44 outputs a high logic level to status pin S3 of 
the parallel port.  The Xilinx software checks for a 
high level on this status pin which indicates that 
power is being supplied to the PCBLIII. 

The Xilinx software also checks for the presence of 
the PCBLIII by looping a signal from parallel port data 
pin D6 back through two of the status pins S5 and 
S7.  Line 49 handles the loop from D6 back to S5.  
The CPLD cannot pass D6 to S7, however, because 
its TDO pin is already attached to S7.  Therefore, the 
shunt on jumper J9 has to be moved to the xi 
position to manually connect D6 and S7. 

Line 51 drives the mode pins of the SpartanII FPGA 
to set it in the slave-serial configuration mode.  This 
doesn’t really do anything since the SpartanII will be 
programmed through its JTAG interface.  Line 52 
holds the /PROGRAM pin of the SpartanII at a high 
logic level to prevent accidental erasure of the FPGA 
configuration.  

Lines 56–58 connect parallel port data pins D2, D1, 
and D0 to the SpartanII TMS, TCK, and TDI JTAG 
pins if D3 is low.  The TMS, TCK and TDI inputs are 
allowed to float when D3 is high. 

Line 61 passes the FPGA JTAG TDO signal back to 
the PC through status pin S4 of the paralle port.  S4 
is driven low when data pin D4 is low. 

The PCBLIII emulator I/O pin assignments for the 
CPLD on the XSA Board are shown in Listing 2. 

Using the Parallel Cable III Emulator 

First, connect the XSA Board to the parallel port of a 
PC through the simple 25-wire cable provided by 
XESS.  Make sure the shunt on jumper J9 is in the xs 
position.  Then download the piiijtag.svf file into the 
XC9572XL CPLD using the GXSLOAD tool from XESS.  
Now move the shunt on J9 to the xi position.  At this 
point, the bitstream downloading portion of the 
PCBLIII emulator is active. 

Next, double-click the Configure Device (iMPACT) icon in 
the Process pane of the WebPACK Project 
Navigator window.  (This discussion assumes you 
already have a SpartanII design synthesized and 
implemented in WebPACK.)  The iMPACT window 
will appear.  Within a few seconds the XSA Board will 
be probed and the JTAG chain consisting of a single 
XC2S100 FPGA will be detected. 

In the lower pane of the iMPACT window, click on the 
xc2s100 object to select this device as the target for 
configuration.  Then select the Operations Program… 
menu item and the Program Options window will 
appear.  Click on the OK button and the bitstream for 
your design will download into the SpartanII FPGA on 
the XSA Board. 

Figure 1: Xilinx Parallel Cable III schematic.  The line numbers of the VHDL code in Listing 1 associated 
with each schematic element are shown. 

Listing 1: VHDL code for the Parallel Cable III emulator. 
library ieee; 1 
use ieee.std_logic_1164.all; 2 
 3 
entity piii is 4 
 port( 5 
  -- parallel port data and status pins 6 
  ppd:    in std_logic_vector(6 downto 0); 7 
  pps:    out std_logic_vector(5 downto 3); 8 
 9 
  -- programmable oscillator 10 
  clk: in std_logic; 11 
   12 
  -- Spartan2 FPGA pins 13 
  S2_tck:   out std_logic; -- driver to Spartan2 JTAG clock 14 
  S2_tms:   out std_logic; -- driver to Spartan2 JTAG mode input 15 
  S2_tdi:   out std_logic; -- driver to Spartan2 JTAG serial data input 16 
  S2_tdo:   in std_logic; -- input from Spartan2 JTAG serial data output 17 
  S2_cclk: out std_logic; -- driver to Spartan2 config clock 18 
  S2_prog_n: out std_logic; -- driver to Spartan2 /PROGRAM pin 19 
  S2_done: in std_logic; -- input from Spartan2 DONE pin 20 
  S2_din:   out std_logic; -- driver to Spartan2 config serial data input 21 
  S2_m:    out std_logic_vector(0 downto 0); -- Spartan2 config mode pins 22 
  S2_clk:   out std_logic; -- clock output to Spartan2 23 
   24 
  fce_n:   out std_logic  -- Flash chip-enable 25 
 ); 26 
end piii; 27 
 28 
architecture arch of piii is 29 
 constant NO: std_logic := '0'; 30 
 constant YES: std_logic := '1'; 31 
 constant LO: std_logic := '0'; 32 
 constant HI: std_logic := '1'; 33 
 constant SLAVE_SERIAL_MODE: std_logic_vector(0 downto 0) := "1"; 34 
begin 35 
 36 
 -- disable Flash chip when Spartan2 is not configured 37 
 fce_n   <= HI when S2_done=NO else 'Z'; 38 
  39 
 -- send the programmable oscillator clock to the Spartan2 40 
 S2_clk   <= clk; 41 
  42 
 -- the XSA power status is sent back through the parallel port status pin 3 43 
 pps(3)  <= HI;  -- tell the PC that the VCC for the XSA board is OK 44 
 -- the cable is detected by sending data through data pin 6 and returning 45 
 -- it on status pins 5 and 7.  Status pin 7 is used by the JTAG TDO 46 
 -- pin of the XC9500 CPLD on the XSA Board, so place a shunt at position "xi" 47 
 -- of jumper J9 to make this connection. 48 
 pps(5)  <= ppd(6); 49 
 50 
 S2_m   <= SLAVE_SERIAL_MODE; -- set Spartan2 config mode pins 51 
 S2_prog_n <= HI;        -- prevent any accidental reconfiguration 52 
 53 
 -- drive the Spartan2 JTAG pins from the parallel port when tristate 54 
 -- control pin (parallel port data pin 3) is low. 55 
 S2_tms  <= ppd(2) when ppd(3)=LO else 'Z'; 56 
 S2_tck  <= ppd(1) when ppd(3)=LO else 'Z'; 57 
 S2_tdi  <= ppd(0) when ppd(3)=LO else 'Z'; 58 
 59 
 -- the JTAG TDO output is sent back through the status pin 60 
 pps(4)  <= S2_tdo when ppd(4)=HI else LO; 61 
end arch; 62 

Listing 2: User-constraint file for CPLD pin assignments. 
# 1 
# pin assignments for the XC9572XL CPLD chip on the XSA Board 2 
# 3 
 4 
# Spartan2 FPGA connections to CPLD 5 
net S2_clk  loc=p42; 6 
net S2_tck  loc=p13; 7 
# net S2_dout  loc=p18; 8 
net S2_tms  loc=p18; 9 
# net S2_din  loc=p2; 10 
# net S2_wr_n  loc=p19; 11 
net S2_tdo  loc=p19; 12 
# net S2_cs_n  loc=p15; 13 
net S2_tdi  loc=p15; 14 
# net S2_init_n  loc=p38; 15 
net S2_done  loc=p40; 16 
net S2_prog_n  loc=p39; 17 
# net S2_cclk  loc=p16; 18 
net S2_m<0>   loc=p36; 19 
# net S2_d<0>  loc=p2; 20 
# net S2_d<1>  loc=p4; 21 
# net S2_d<2>  loc=p5; 22 
# net S2_d<3>  loc=p6; 23 
# net S2_d<4>  loc=p7; 24 
# net S2_d<5>  loc=p8; 25 
# net S2_d<6>  loc=p9; 26 
# net S2_d<7>  loc=p10; 27 
 28 
# Flash RAM 29 
# net fd<0>  loc=p2; 30 
# net fd<1>  loc=p4; 31 
# net fd<2>  loc=p5; 32 
# net fd<3>  loc=p6; 33 
# net fd<4>  loc=p7; 34 
# net fd<5>  loc=p8; 35 
# net fd<6>  loc=p9; 36 
# net fd<7>  loc=p10; 37 
# net fa<0>  loc=p1; 38 
# net fa<1>  loc=p64; 39 
# net fa<2>  loc=p63; 40 
# net fa<3>  loc=p62; 41 
# net fa<4>  loc=p61; 42 
# net fa<5>  loc=p60; 43 
# net fa<6>  loc=p59; 44 
# net fa<7>  loc=p58; 45 
# net fa<8>  loc=p45; 46 
# net fa<9>  loc=p44; 47 
# net fa<10>  loc=p57; 48 
# net fa<11>  loc=p43; 49 
# net fa<12>  loc=p56; 50 
# net fa<13>  loc=p46; 51 
# net fa<14>  loc=p47; 52 
# net fa<15>  loc=p52; 53 
# net fa<16>  loc=p51; 54 
# net fa<17>  loc=p48; 55 
# net frst_n  loc=p50;# Flash reset 56 
# net foe_n  loc=p12;# Flash output-enable 57 
# net fwe_n  loc=p49;# Flash write-enable 58 
net fce_n  loc=p11;# Flash chip-enable 59 
 60 
# DIP and pushbutton switches 61 
# net dipsw<1>  loc=p47; 62 

# net dipsw<2>  loc=p52; 63 
# net dipsw<3>  loc=p51; 64 
# net dipsw<4>  loc=p48; 65 
 66 
# 7-segment LEDs 67 
# net s<0>  loc=p10; 68 
# net s<1>  loc=p2; 69 
# net s<2>  loc=p9; 70 
# net s<3>  loc=p8; 71 
# net s<4>  loc=p5; 72 
# net s<5>  loc=p7; 73 
# net s<6>  loc=p6; 74 
# net dp loc=p4; 75 
 76 
# programmable oscillator 77 
net clk   loc=p17; 78 
 79 
# parallel port 80 
net ppd<0>      loc=p33; 81 
net ppd<1>      loc=p32; 82 
net ppd<2>  loc=p31; 83 
net ppd<3>  loc=p27; 84 
net ppd<4>  loc=p25; 85 
# net ppd<5>  loc=p24; 86 
net ppd<6>  loc=p23; 87 
# net ppd<7>  loc=p22; 88 
net pps<3>  loc=p34; 89 
net pps<4>  loc=p20; 90 
net pps<5>  loc=p35; 91 
 92 

